[介護ハック]経菅栄養自動加圧送出装置を作った～

##だれもが通る親の介護が始まりました
うちの父は嚥下がうまく出来なくて、大部分を経菅栄養で補っています。
液状の栄養剤を管に流して胃に送り込むのですが、液体だと下痢になる場合が多く、最近では粘り気のある半固形の栄養剤を使うようになっています。
![alt](https://www.otsukakj.jp/med_nutrition/dikj/upload/rnh05a1ab.jpg)

半固形の栄養剤は液体のようにぶる下げて重力では流せないので、シリンジで注入したり、加圧バッグでポンピングして注入します。
![racole.jpg](http://www.peg.or.jp/lecture/peg/product/nipro_bag/00.jpg)
[加圧バッグでの注入動画](https://www.otsukakj.jp/lib/image/healthcare/home_nutrition/kaatsubag.mp4)
##安定した排便に向けた調整
うちでは半固形栄養剤だとだんだん便秘になってしまいました。
粘度のある液体は摩擦があり腸内をゆっくり通過するため、栄養や水分が吸収されやすくなります。便を軟らかくするには、粘度を下げて腸通過を早め、食物繊維を増やし大腸まで水分を保てるようにします。
軟らかくなるように[食物繊維](https://my-best.com/3831)や水分や[液体栄養剤](http://www.peg.or.jp/lecture/enteral_nutrition/product/otsuka_racol-nf/index.html)を混ぜています。
![alt](https://www.otsukakj.jp/med_nutrition/dikj/upload/rnf05a1ab_1.jpg)
栄養の粘度が下がるので、加圧バッグの空気圧の調整を細めにしないと注入速度が速くなりすぎて胃から食道に逆流するリスクが高まります。その為、加圧バッグを秤に載せて微妙な減り具合をみながら手動でポンピングしていました。
チューブを流れる栄養の速さが見えにくく、秤の細かい目盛りを凝視しながら、小まめにポンピングして、十数分間で終了するように調整するのは、高齢の母にはつらい作業でした。
長期にわたるだろう介護の負担をできるだけ軽減できないかと思い、自動加圧装置作成に至りました。
##経菅栄養自動加圧送出装置の作成
手動で行っていた重量測定とポンピングを自動化しようと、Arduinoで秤センサーを読み取りエアポンプを制御する事にしました。
![66369392_2800103206726738_7158757412009148416_n.jpg](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/106489/ff640304-a9dc-993b-2deb-d8c675d76c63.jpeg)
秤センサーとプッシュスイッチがデジタル入力、エアポンプとLEDインジケーターがデジタル出力になります。電源アダプターは、9.5V 900mAのトランスタイプのレギュレータ無しを使いました。

![injectorsch.jpg](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/106489/8550918b-2567-20e1-b9f8-f049fd93301d.jpeg)
###秤センサー
対象物は700gぐらいなので、1kgぐらいの秤センサーを検索してみるとAmazon発送で販売している中華業者で[5kgの物￥1300が見つかりました](https://www.amazon.co.jp/HX711%E3%83%AD%E3%83%BC%E3%83%89%E3%82%BB%E3%83%AB%E3%83%A2%E3%82%B8%E3%83%A5%E3%83%BC%E3%83%AB-24%E4%BD%8D%E7%B2%BE%E5%BA%A6AD%E3%83%A2%E3%82%B8%E3%83%A5%E3%83%BC%E3%83%AB-%E5%9C%A7%E5%8A%9B%E3%82%BB%E3%83%B3%E3%82%B5%E3%83%BC%E3%83%A2%E3%82%B8%E3%83%A5%E3%83%BC%E3%83%AB-Arduino-%E9%87%8D%E9%87%8F%E3%82%BB%E3%83%B3%E3%82%B55KG/dp/B07JL7NP3F/ref=sr_1_2?__mk_ja_JP=%E3%82%AB%E3%82%BF%E3%82%AB%E3%83%8A&keywords=5kg+%E3%82%BB%E3%83%B3%E3%82%B5%E3%83%BC+arduino&qid=1562997657&s=gateway&sr=8-2)。分解能24ビットという事で1g未満でも計測できそうなので入手してみました
![loadcell](https://images-na.ssl-images-amazon.com/images/I/61qE%2B6hhiBL._SX425_.jpg) 
中華部品が翌日届くというのは新鮮な驚きがありました～
###Arduino,PowerMOSFET
Arduinoは手持ちの[安い中華Uno互換機￥281](https://www.aliexpress.com/item/32803501325.html?spm=2114.search0104.3.1.45812f6aqbi2lS&ws_ab_test=searchweb0_0%2Csearchweb201602_2_10065_10068_319_10059_10884_317_10887_10696_321_322_10084_453_10083_454_10103_10618_10307_537_536%2Csearchweb201603_53%2CppcSwitch_0&algo_expid=dd31b010-65eb-463e-b3bb-46324d9c14c6-0&algo_pvid=dd31b010-65eb-463e-b3bb-46324d9c14c6&transAbTest=ae803_3)、エアポンプ駆動用の[PowerMOSFET ￥50](http://akizukidenshi.com/catalog/g/gI-08452/)と[ヒートシンク￥30](http://akizukidenshi.com/catalog/g/gP-05052/)は秋月でを入手しました
###エアポンプ
どのくらいの空気圧が必要なのかよくわからなかったので、
まずは自動車タイヤの[電動空気入れを試してみました](https://www.amazon.co.jp/150PSI%E3%82%A8%E3%82%A2%E3%83%BC%E3%82%B3%E3%83%B3%E3%83%97%E3%83%AC%E3%83%83%E3%82%B5%E3%83%BC-%E8%87%AA%E5%8B%95%E8%BB%8A-%E3%83%90%E3%82%A4%E3%82%AF-%E3%82%A4%E3%83%B3%E3%83%95%E3%83%AC%E3%83%BC%E3%82%BF%E3%83%96%E3%83%AB%E3%83%95%E3%83%AD%E3%83%BC-%E3%82%AA%E3%83%BC%E3%83%88%E3%82%B9%E3%83%88%E3%83%83%E3%83%97%E6%A9%9F%E8%83%BD%E3%83%BB%E3%83%87%E3%82%B8%E3%82%BF%E3%83%AB%E8%A1%A8%E7%A4%BA%E3%83%BBLED%E3%83%A9%E3%82%A4%E3%83%88%E4%BB%98/dp/B07KTZ991S/ref=sr_1_50_sspa?__mk_ja_JP=%E3%82%AB%E3%82%BF%E3%82%AB%E3%83%8A&keywords=12v+%E3%82%A8%E3%82%A2%E3%83%BC+%E8%87%AA%E5%8B%95%E8%BB%8A&qid=1562998773&s=gateway&sr=8-50-spons&psc=1)が、騒音が大きくて空気圧が高すぎました。
次にAmazon発送で販売している[中華業者の物￥1500が見つかりました](https://www.amazon.co.jp/DC12V-%E3%83%9E%E3%82%A4%E3%82%AF%E3%83%AD%E7%9C%9F%E7%A9%BA%E3%83%9D%E3%83%B3%E3%83%97-%E9%9B%BB%E6%B0%97%E3%83%9D%E3%83%B3%E3%83%97-%E3%83%9F%E3%83%8B%E3%82%A8%E3%82%A2%E3%83%9D%E3%83%B3%E3%83%97%E3%83%9D%E3%83%B3%E3%83%97-%E5%90%84%E7%A8%AE%E5%A0%B4%E5%90%88%E3%81%AB%E5%AF%BE%E5%BF%9C/dp/B07KPFC2Z2/ref=pd_sbs_60_4/357-9812687-4461058?_encoding=UTF8&pd_rd_i=B07KPFC2Z2&pd_rd_r=025d04ab-a539-11e9-8924-833d5d723e91&pd_rd_w=hOQFi&pd_rd_wg=nESnl&pf_rd_p=ad2ea29d-ea11-483c-9db2-6b5875bb9b73&pf_rd_r=G3QEXMD49YY3AT0D91QG&psc=1&refRID=G3QEXMD49YY3AT0D91QG)。空気圧のスペックが50kpaということで、ちょうどよい空気圧でしたので採用となりました。

空いた菓子箱にマイコンとエアポンプを収めて、上ふたにLEDとスイッチを配置しました。
外付け部品は少ないのでArduinoに部品直付けで配線してしまいました。
![inject2.jpg](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/106489/ec94dfcf-5645-1a43-3ed6-fe92e6de8df3.jpeg)
#プログラミング
スケッチは、秤センサーの[HX711ライブラリー](https://github.com/bogde/HX711)を利用しています。
以下プログラムの概要です
-スタートスイッチが押されるまで待ちます。
-まず加圧バッグがふくらんで栄養が流れ始めて重量が6g減るまでポンプを回します。
-初期重量から終了重量まで、120ステップ/8秒毎に現在の重量を測って、目標重量より重かった場合にポンプを0.2秒回します。
-5個のLEDインジケーターは、状態毎に点灯パターンを変えています。

```HX711_AUTO_PEG_INJECTOR.INO
#include "HX711.h"

// HX711 circuit wiring
const int LOADCELL_DOUT_PIN = 13;
const int LOADCELL_SCK_PIN = 12;

const int PUMP_PIN = 11;
const int PUSHSW_PIN = 10;
const int LED3_PIN = 3;

const long fin= 300000;

int i=0;
long total,start,reading;

HX711 scale;

// scale weight function
long weight() {
    while (!scale.is_ready()) {
      Serial.println("HX711 not ready.");
      delay(200);
    }
    long w = scale.read();
    Serial.print("HX711 reading: ");
    Serial.println(w);
    return w;
  }
  
void setup() {
  Serial.begin(57600);
  pinMode(PUSHSW_PIN, INPUT_PULLUP);

  pinMode(PUMP_PIN, OUTPUT);
  digitalWrite(PUMP_PIN, LOW);
  
  for(i=LED3_PIN; i<8; i++) {
    pinMode(i, OUTPUT);
    digitalWrite(i, LOW);
  }
  scale.begin(LOADCELL_DOUT_PIN, LOADCELL_SCK_PIN);
}


void loop() {
  // wait for Switch pressed
  while ( digitalRead(PUSHSW_PIN) == LOW) delay(100);

  while ( digitalRead(PUSHSW_PIN) == HIGH) {
    digitalWrite(i-1, LOW);
      if(i==8) i=LED3_PIN;
    digitalWrite(i++, HIGH);    
    delay(100);
  }
//DEBUG  while(1) weight();
// scale 5 times average   
  start=0;
  for(i=0; i<5; i++) start+=weight();
  start /= 5;
  Serial.print("HX711 starting: ");
  Serial.println(start);
// initial pumping
  i=0;
  reading = weight();
  // until 6g decrease
  while(reading>(start-1500)) {
    Serial.println(i);
// turn PUMP on:
    digitalWrite(PUMP_PIN, HIGH);
      delay(500);
    digitalWrite(PUMP_PIN, LOW);
      delay(100);
    reading = weight();
    if( i>27 ) {
  // OVER PRESSURE ERROR 
      while (1) {
      for(i=LED3_PIN; i<8; i++)
        digitalWrite(i, LOW);
      delay(100);
      for(i=LED3_PIN; i<8; i++)
        digitalWrite(i, HIGH);
      delay(100);
      if ( digitalRead(PUSHSW_PIN) == LOW) return;
      }
    }
    i++;
  }

  // turn PUMP off:
  digitalWrite(PUMP_PIN,LOW );
  for(i=LED3_PIN; i<8; i++)
    digitalWrite(i, LOW);
 // Start Injecting     
  for(i=119; i>0; i--) {

     digitalWrite(LED3_PIN+(i/24), HIGH);

    reading = weight();
    Serial.print(i);
    // compare weight with reference
    if( reading > fin+((start-fin) * i / 120)) {
      // turn PUMP on:
      digitalWrite(PUMP_PIN, HIGH);
      delay(200);
      // turn PUMP off:
      digitalWrite(PUMP_PIN,LOW );
    }
    // 8sec * 120 = 16min to complete
   for(int j=0;j<8;j++) {
    // 1000msec * 8 = 8000msec
    // Blink LED with iter period
     digitalWrite(LED3_PIN+(i/24), LOW);
     delay(500+((i%24)*20)); 
     digitalWrite(LED3_PIN+(i/24), HIGH);
     delay(500-((i%24)*20)); 
   }

  }
  // finish work
  while (1) {
  // blink LED
    for(i=LED3_PIN; i<8; i++) {
      digitalWrite(i, LOW);
  // Push switch to restart
      if ( digitalRead(PUSHSW_PIN) == LOW) return;
    }
    delay(1500);
    for(i=LED3_PIN; i<8; i++) {
      digitalWrite(i, HIGH);
      if ( digitalRead(PUSHSW_PIN) == LOW) return;
    }
    delay(1500);
    }

//  Serial.println(digitalRead(PUSHSW_PIN));
  }
```
#運用に向けて
デバッグはシリアルポートに状態を垂れ流して、PCから観測しました。
自分で何度か試して問題が無い事を確認してから、
母親が実際に運用してみると、異常系の対応がいくつか必要になり、
重要な異常系処理コードを追加しました。
運用途中での異常系からの復帰はあきらめ、開始前のチェックポイントをスタートスイッチの上に書いて異常発生を防ぐ事で対応としました。
![injector3.jpg](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/106489/630bcc3c-07d2-d2e9-8edd-a99a5086dad3.jpeg)
またハード等の異常でポンプが止まらなくなった場合には、空気栓を開いて電源を落とす事を周知しました。

---
一週間ほど現場で順調に運用できています。経管栄養作業の負担が身体的にも精神的にも軽減されて、母に余裕すら見えてきた今日この頃です